

```

1  finish                      ! kilépés minden process-ből
2  /clear                      ! memória törlése
3
4  /cwd, 'D:\Google_Drive\01_education\07_VEA\19-20-1\APDL\feladat2-eromero_cella'
5  *abbr, START, /input, SA-eromero_cella-basic, ans
6
7  ! ADATBEVITEL !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
8
9  ! Geometriai adatok
10 l=80                        ! az erőmérő cella hossza
11 a=12                        ! az erőmérő cella magassága
12 s=9                         ! furatok középpontjainak távolsága
13 d=10                        ! furatok átmérője
14 b=8                         ! üreg kibővítése egy négyzettel (élhossz: 8mm)
15
16 ! Anyagtulajdonságok
17 E=2e5                       ! rugalmassági modulusz
18 nu=0.3                      ! Poisson tényező
19
20 ! Peremfeltételek
21 l0=20                       ! megfogás hossza
22 l1=30                       ! terhelés hossza
23 n=1                         ! terhelési eset
24
25 ! ELŐKÉSZÍTÉS !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
26
27 /prep7                      ! belépés a preprocesszorba
28
29 ! Geometria megrajzolása
30 blc4,0,0,l,a                ! téglalap rajzolása
31 cm,cella,area               ! téglalap elnevezése cellának
32
33 cyl4,l/2-s/2,a/2,d/2        ! baloldali furat megadása
34 cyl4,l/2+s/2,a/2,d/2        ! jobboldali furat megadása
35 blc5,l/2,a/2,b,b            ! üreg kibővítése
36 cmsel,u,cella               ! első téglalap (cella) elrejtése az aktuális elemek közül
37 cm,ureg,area                ! ami most látható (2 furat + négyzet) elnevezése ureg-nek
38
39 allsel                      ! minden kijelölése (újra minden látható lesz)
40 asba,cella,ureg             ! az ureg kivonása a cellából
41
42 !Geometria előkészítése hálózashoz:
43 lsel, none                   ! Összes vonal elrejtése
44 k,100,l/2-s/2-d/2-0.1*d/2,0 ! Csomópontok definiálása a felület particionálásához
45 k,101,l/2-s/2-d/2-0.1*d/2,a
46 k,102,l/2+s/2+d/2+0.1*d/2,a
47 k,103,l/2+s/2+d/2+0.1*d/2,0
48 l,100,101                   ! Vonalak definiálása a felület particionálásához
49 l,102,103
50
51 cm,cut,line                  ! aktuális vonalak elnevezése cut-nak
52 asbl,all,cut                ! cut kivonása az egész felületből = particionálás
53
54 asel,s,loc,x,l/2            ! a középső felületrész kijelölése
55 cm,uregresz,area            ! a kijelölt rész elnevezése uregresz-nek
56
57 asel,inve                   ! az üregrészen kívüli részek kijelölése
58 cm,kulsoresz,area           ! a kijelölés elnevezése külső résznek
59
60 allsel
61
62 ! Geometria előkészítése peremfeltételekhez
63 lsel,s,loc,y,0               ! az alsó vonal kijelölése
64 ldiv,all,l0/l               ! a kijelölt vonal felosztása 20 és 60 mm-es szakaszokra
65
66 lsel,s,loc,y,a               ! a felső vonal kiválasztása
67 ldiv,all,l1/l               ! a kijelölt vonal felosztása 30 és 50 mm-es szakaszokra
68
69 lsel,r,loc,x,l-l1,l         ! a 30 mm-es szakasz kijelölése
70 ldiv,all,,,3                 ! a kijelölt szakasz felosztása 3 egyenlő részre
71

```

```

72 ! Anyagtulajdonságok megadása:
73 mp,ex,1,E ! rugalmassági modulusz
74 mp,nuxy,1,nu ! Poisson tényező
75
76 ! Végeelem háló beállítása, elkészítése
77 et,1,182,,,2 ! elemtípus: 182: 2D, 2:SA
78 mshape,1,2d ! elem alakja (0: négyszög, 1: háromszög)
79 esize,0.5 ! elemméret: 0.5 mm
80 amesh, uregresz ! az uregresz behálózása
81 !desize,1,1,9999,,,0.5,5
82 esize 5 ! elemmeret: 5
83 amesh, kulsoresz ! a kulsoresz behalozasa
84
85 ! Peremfeltételek megadása
86 lsel,s,loc,y,0 ! a befogás helyének kijelölése
87 lsel,r,loc,x,0,10
88 nsll,s,1 ! összes csomópont kijelölése az aktív vonalon
89
90 d,all,ux,0 ! A kijelölt csomópontok megfogása x irányban
91 d,all,uy,0 ! A kijelölt csomópontok megfogása y irányban
92
93 lsel,s,loc,y,a ! a megadott terhelési szakasz kijelölése
94 lsel,r,loc,x,1-11+(n-1)*11/3,1-11+n*11/3
95 sfl,all,pres,1 ! terhelés elhelyezése (p=1N/mm^2)
96
97 allsel ! minden kijelölése (újra minden látható lesz)
98
99 finish
100
101 ! SZÁMÍTÁS !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
102 /sol ! belépés a solver-be
103 solve ! számítás elvégzése
104 finish ! kilépés a solver-ből
105
106
107 ! KIÉRTÉKELÉS !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
108 /post1 ! belépés a postprocessor-ba
109
110 plnsol,u,sum,2,1.0 ! elmozdulásmező szemléltetése (2: eredeti alak is, 1.0:
nagyítás)
111
112 /wait,4 ! várakozás 4 sec-ig
113
114 plnsol,epel,x,2,1.0 ! epsilon_x fajlagos nyúlás szemléltetése
115
116 /wait,4 ! várakozás 4 sec-ig
117
118 plnsol,s,eqv,2,1.0 ! redukált feszültség szemléltetése
119
120 /status,all
121

```